

*Scuola Media «Maurizio Sacchi»
Mantova - Classi terze A-E*

abbondanti zero

Dizionario dei numeri naturali

*In collaborazione con
Assessorato alla Pubblica Istruzione
Comune di Mantova*

*Le cifre,
lungi dall'essere simboli secchi e aridi
che molti denunciano come armi e vettori
della nostra civiltà tecnologica,
sono state in ogni tempo anche supporto
di sogno, di trascendenza,
di speculazione metafisica,
nonché materiale di letteratura,
sonde dell'incerto avvenire o
almeno del desiderio di predirlo.
Le cifre sono una sostanza poetica.*

Georges Ifrah, Storia Universale dei Numeri

UNO DEI POSSIBILI FILI D'ARIANNA

Somma Divisori	Scomposizione	Analisi Divisori
1 Abbondanti	1 Coprimi	1 Composti
2 Amici	2 Eterogenei	2 Dispari
3 Deficienti	3 Omogenei	3 Disparimentipari
4 Duffyniani	4 Primi	4 Pari
5 Perfetti		5 Parimentidispari
6 Quasiperfetti		6 Parimentipari
7 Socievoli		
8 Superabbondanti		

Senza manipolare le cifre	Manipolando le cifre
1 Bifattoriali	1 Autoriproducentesi
2 Casuali	2 Bifronti
3 Cubici	3 Cherry
4 Fattoriali	4 Narcisi
5 Fibonacci	5 Omar
6 Planici	6 Palindromi
7 Quadrati	
8 Triangolari	

Numeri e parole	La magia	Aritmogeometria
1 Eteroletterali	1 Magici	1 Artful
2 Numerali	2 Uno	2 Pitagorici
3 Onesti	3 Venti	
4 Stringa	4 Zero	

Proprietà curiose	Necessità alfabetiche	
1 Automorfi	1 Guinnessiani	4 lotto
2 Biperiodici	2 Hotel	5 Romani
3 Conservativi	3 Interi	

ABBONDANTI- Un numero intero positivo è abbondante se è minore della somma dei suoi divisori propri.

SDP(12): $1+2+3+4+6=16$; $12 < 16$; allora 12 è abbondante

I primi numeri abbondanti: 12 - 18 - 20 - 24 - 30 - 36

Osservazioni

- * Nei primi 1000 numeri naturali ci sono 246 abbondanti e, a parte il 945, sono tutti pari
- * Dato un numero abbondante il successivo lo si trova o dopo 2 o dopo 4 o dopo 6 numeri
- * Tutti i multipli di un abbondante sono abbondanti
- * Tra un abbondante e l'altro c'è sempre un numero primo
- * Tra due numeri primi gemelli (vedi PRIMI) c'è sempre un abbondante
- * Tutti i multipli dei numeri perfetti (vedi *) sono abbondanti

*L'oceano
si affaccia sulla spiaggia...
Abbondante atlantico*

Programma

```
100 'I Numeri Abbondanti
110 CLS : KEY OFF
120 INPUT "Limite inferiore ";A
130 PRINT
140 INPUT "Limite superiore ";B
150 CLS
160 FOR N=A TO B
170   SD=1
180   FOR K=2 TO SQR(N)
190     Q=N/K
200     IF Q=K THEN SD=SD+K : GOTO 220
210     IF Q=INT(Q) THEN SD=SD+K+Q
220   NEXT K
230   IF N<SD THEN T=T+1 : PRINT T,N
240   IF T/20=T\20 AND C<>T THEN PRINT : INPUT "
    Premi INVIO",D : CLS : C=T
250 NEXT N
260 PRINT : PRINT "Totale Abbondanti: ";T
```


270 LOCATE 23,1
280 END

AMICI - Due numeri sono amici quando ciascuno è uguale alla somma dei divisori propri dell'altro.

SDP(220): $1+2+4+5+10+11+20+22+44+55+110=284$

SDP(284): $1+2+4+71+142=220$

Allora 284 e 220 sono numeri amici

Una volta Pitagora disse: "Un amico è l'altro io, come lo sono il 220 e il 284".

I commentatori biblici sottolineavano il fatto che le capre date da Giacobbe ad Esaù erano 220, amico di 284.

Nel Medioevo, questa coppia di numeri veniva usata nella preparazione degli oroscopi e si pensava che, se incisi su un talismano, portassero fortuna e amore.

Secondo un'antica usanza araba gli innamorati si scambiano dei dolci sui quali sono segnati i due numeri magici.

Questa coppia di numeri, però, non è l'unica che si conosca. I matematici ne hanno scoperte più di 1000 anche se non esiste una formula generale che permetta di ricavarle e non si sa ancora se siano infinite.

Le coppie di numeri amici inferiori a 100.000:

(220-284) - (1.184-1.210) - (2.620-2.924) - (5.020-5.564) -
(6.232-6.368) - (10.744-10.856) - (12.285-14.595) - (17.296-
18.416) - (63.020-76.084) - (66.928-66.992) -
(67.095-71.145) - (69.615-87.633) - (79.750-88.730)

Osservazioni

- * I numeri amici, finora conosciuti, sono costituiti da due numeri pari o, più raramente, da due numeri dispari
Non si conoscono coppie di parità opposta
- * Le coppie di amici dispari sono numeri multipli di tre
- * Se un numero della coppia è abbondante (vedi *) l'altro è deficiente (vedi *)
- * I numeri primi (vedi *) sicuramente non hanno amici
- * I numeri perfetti (vedi *) sono amici di se stessi

Nel bosco fitto

*anemone si apre -
Amici nella cascata*

Programma

```
100 Test per le coppie di Amici
110 CLS : KEY OFF
120 INPUT "Scrivi un numero: ",N
130 CLS
140 A=N : G=1
150 WHILE A<>SD AND G<4
160     SD=1
170     FOR K=2 TO SQR(N)
180         Q=N/K
190         IF Q=INT(Q) THEN SD=SD+K+Q
200     NEXT K
210     F(G)=SD : G=G+1
220     N=SD
230 WEND
240 IF G>3 THEN PRINT A;" non ha amici" :
    ELSE PRINT A;" è amico di ";F(1)
250 LOCATE 23,1
260 END
```

ARTFUL - Dall'inglese art=arte e full=pieno. Fatto ad arte.

Sono quei numeri interi che possono essere area di un triangolo.

Es. 12 è artful perché area di un triangolo di lati 5; 5; 6

Osservazioni

- * Nessun artful è area di un triangolo equilatero
 - * I triangoli rettangoli, i cui lati sono terne pitagoriche (vedi PITAGORICI), generano numeri artful. Infatti, almeno uno dei due cateti è sempre pari
 - * I triangoli isosceli, formati da due triangoli rettangoli pitagorici, simmetrici rispetto ad un cateto, generano numeri artful.
- Se la terna è 5, 12, 13 allora il triangolo di lati 10, 13, 13 genera il numero 60

- * I triangoli scaleni formati da due triangoli rettangoli pitagorici che abbiano un cateto in comune, generano numeri artful. Se le due terne sono 14, 48, 50 e 20, 48, 52 allora il triangolo scaleno di lati 34, 50, 52 genera il numero 816

*Un salice verde splende
Artful:
misteriosa nevicata*

AUTOMORFI - Dal greco autòs=stesso e morphé=forma
Sono quei numeri che appaiono nella parte finale del proprio quadrato. Es. $76 \cdot 76 = 5776$

Osservazioni

- * Nei primi 1000 numeri se ne trovano 8: 0, 1, 5, 6, 25, 76, 376, 625
- * In generale solo i numeri che hanno il 5 o il 6 come cifre finali possono essere automorfi

*La farfalla bianca si desta -
Automorfo:
delicato vento che si spegne*

Programma

```
100 'I numeri Automorfi
110 CLS : KEY OFF
120 INPUT "Limite inferiore: ",A :PRINT
140 INPUT "Limite superiore: ",B
150 CLS
160 FOR K=A TO B
170   K$=STR$(K)
180   L=LEN(K$)-1
190   P=K*K
200   P$=STR$(P)
210   UC$=RIGHT$(P$,L)
220   UC=VAL(UC$)
230   IF UC=K THEN PRINT K,P
240 NEXT K
250 LOCATE 23,1
```


AUTORIPRODUCENTESI - Sono quei numeri con le cifre in ordine decrescente che, sottratti al loro bifronte (vedi *), danno come risultato un numero con le stesse cifre.

Numeri di due cifre: nessun esempio

Numeri di tre cifre: $954-459=495$

Numeri di quattro cifre: $7641-1467=6174$

Osservazioni

- * Tutti i numeri con le tre cifre consecutive in ordine decrescente danno come risultato 198
Es. $654-456=198$
- * Tutti quei numeri che hanno le prime due cifre consecutive e la terza distanziata di due dalla seconda, danno come risultato 297
Es. $653-356=297$
- * Tutti quei numeri che hanno le prime due cifre consecutive e la terza distanziata di tre dalla seconda, danno come risultato 396
Es. $874-478=396$
- * Tutti quei numeri che hanno le cifre pari (o le cifre dispari) consecutive danno come risultato 396
Es. $864-468=396$; $531-135=396$
- * Ma, in generale, si può osservare che qualunque sottrazione di questo tipo ha come cifra centrale il 9 e, che la somma della prima e dell'ultima è ancora 9. Ciò significa che ci sono solo 10 possibili risultati: 099 ; 198 ; 297 ; 396 ; 495 e i loro bifronti.

Ora, se con ognuno di questi numeri si ordinano le cifre in modo decrescente e si sottrae il loro bifronte, ripetendo questo procedimento alcune volte, si arriverà sicuramente a 495

E, 495 viene chiamato il POLO dei numeri di tre cifre.

Es.(099) $990-099=891$; $981-189=792$; $972-279=693$;
 $963-369=594$; $954-459=495$

Quindi, se applichiamo questo procedimento a qualunque numero di tre cifre, prima o poi, arriveremo a 495

*Nei numeri di due cifre la sottrazione produce sempre numeri che hanno la somma delle cifre uguale a 9 e cioè 09 ; 18 ; 27 ; 36 ; 45 e i loro bifronti.

Ma questi numeri non hanno il loro POLO perché si riproducono a vicenda

Es.(72) $72-27=45$; $54-45=09$; $90-09=81$; $81-18=63$;
 $63-36=27$; $72-27=45$

*I numeri di quattro cifre hanno come POLO 6174

*Foresta profumata -
stanotte un ramo si muove
mormora l'Autoriproducentesi*

Programma

```
100 'Il Polo di un numero
110 CLS : KEY OFF
120 WHILE N<102 OR N>999
130   INPUT"Scrivi un numero di tre cifre: ",N
140   CLS
150 WEND
160 GOSUB 220
170 R=NO-BF
180 PRINT R
190 IF R<>495 THEN N=R : GOTO 160
200 LOCATE 23,1
210 END
220 'Ordinamento cifre e loro ribaltamento
230 NO$="" : BF$=""
240 T=T+1
250 PRINT T,
260 N$=STR$(N)
270 L=LEN(N$)
280 FOR K=1 TO L
290   C$(K)=MID$(N$,K,1)
300 NEXT K
310 FOR K=2 TO L-1
320   FOR J=K+1 TO L
330     IF C$(K) < C$(J) THEN SWAP C$(K),C$(J)
340   NEXT J
350 NEXT K
```



```

360 FOR K=2 TO L
370  NO$=NO$+C$(K)
380 NEXT K
390 NO=VAL(NO$)
400 PRINT NO;"-";
410 NO$=STR$(NO)
420 L=LEN(NO$)
430 FOR K=L TO 1 STEP -1
440  BF$=BF$+MID$(NO$,K,1)
450 NEXT K
460 BF=VAL(BF$)
470 PRINT BF;"=";
480 RETURN

```

BIFATTORIALI - Sono il prodotto di tutti i numeri inferiori a n e di ragione 2 (vedi anche FATTORIALI).

Es. $6!! = 2 \cdot 4 \cdot 6 = 48$

$7!! = 1 \cdot 3 \cdot 5 \cdot 7 = 105$

Osservazioni

- * I bifattoriali di un numero pari sono pari
- * I bifattoriali di un numero dispari sono dispari
- * Probabilmente non hanno significato $0!!$ e $1!!$

*Un uragano solitario splende
fiorisce il Bifattoriale
nebbioso salice...*

Programma

```

100 'I numeri Bifattoriali
110 CLS : KEY OFF
120 INPUT "Scrivi un numero >1: ";N
130 IF N/2=N\2 THEN BF=2 ELSE BF=1
140 CLS
150 FOR K=BF TO N STEP 2
160  BF=BF*K
170 NEXT K
180 PRINT "Il bifattoriale di";N;"è:";BF
190 LOCATE 23,1
200 END

```

BIFRONTI - Un numero intero è detto bifrante se, letto da sinistra a destra e da destra a sinistra, cambia il suo valore (vedi anche PALINDROMI).

Es. 1647 e 7461 sono bifranti

Osservazioni

- * La maggior parte dei numeri è bifrante
- * Se consideriamo i numerali (vedi *) l'unico bifrante è UNO che diventa ONU, in Esperanto OK diventa KO, in Inglese TEN diventa NET

*Nella profonda notte
solitario...
palpita il Bifrante*

Programma

```
100 'Generazione numeri Bifranti
110 CLS : KEY OFF
120 INPUT "Scrivi un numero: ";N
130 CLS
140 N$=STR$(N)
150 L=LEN(N$)
160 FOR K=L TO 2 STEP -1
170   B$=B$+MID$(N$,K,1)
180 NEXT K
190 B=VAL(B$)
200 PRINT "il bifrante di ";N;"è";B
210 END
```

BIPERIODICI - Sono quei numeri formati da due gruppi uguali di cifre.

Es. 473473 ; 1234512345

Osservazioni

- * I biperiodici si ottengono moltiplicando il primogruppo per 1001
- * Siccome 1001 è divisibile per 7, 11, 13 allora lo saranno anche tutti i biperiodici

Oggi l'orchidea

*si muove nella pianura
tace il Biperiodico*

```
Programma
100 'I numeri Biperiodici
110 CLS : KEY OFF
120 INPUT "Scrivi un numero: ",N
130 PRINT
140 BP=N*1001
150 PRINT BP
160 LOCATE 23,1
170 END
```

CASUALI - Sono quei numeri la cui successione viene generata in maniera casuale. Sembra che non sia molto facile costruire manualmente queste successioni, perché, inconsapevolmente, siamo portati ad inserire elementi soggettivi.

Con l'aiuto di un elaboratore, questi elementi si possono, quasi del tutto, eliminare e le successioni ottenute vengono definite pseudo-casuali.

Osservazioni

- * Come si collezionano francobolli o monete o pipe, così è interessante, e forse meno dispendioso, collezionare cifre casuali.
- * Prendiamo le dieci cifre del nostro sistema di numerazione e le facciamo generare da un computer. L'elaboratore terminerà la successione quando le avrà generate tutte.
- * Quante saranno le cifre di una successione?
Ogni cifra ha 10 possibilità di verificarsi, quindi:
 $10 \cdot (1 + 1/2 + 1/3 + 1/4 + 1/5 + 1/6 + 1/7 + 1/8 + 1/9 + 1/10) = 29,3$
circa
- * Abbiamo collezionato, più volte, cinque successioni di cifre casuali e la media è sempre stata intorno al 30

*Per l'usignolo amato
la memoria si muove
i Casuali ascoltano*


```

Programma
100 'Collezioni di cifre Casuali
110 CLS : KEY OFF
120 DIM C(50)
130 INPUT"Quante collezioni ";Q
140 CLS
150 FOR J=1 TO Q
160 PRINT"Collezione: ";J
170   T=0 : F=0
180   FOR K=0 TO 9
190     C(K)=0
200   NEXT K
210   RANDOMIZE TIMER
220   WHILE F=0
230     N=INT(RND*10)
240     IF POS(0)>68 THEN PRINT
250     PRINT N;
260     T=T+1
270     C(N)=1
280     FOR K=0 TO 9
290       IF C(K)=0 THEN 320
300     NEXT K
310     F=1
320   WEND
330   PRINT" (";T;")"
340   PRINT
350   SC=SC+T
360   IF J/5=J\5 AND J<>Q THEN LOCATE 22,30 :
     INPUT"PREMI INVIO ",D : CLS
370 NEXT J
380 LOCATE 22,1 : PRINT"Media: ";SC/Q
390 END

```

CHERRY - Dall'inglese cherry=ciliegia, ma non sappiamo perché si chiamano così. Forse perché 'una ciliegia tira l'altra'?

Sono delle sequenze che, a partire da un numero arbitrario, hanno come termine successivo la somma dei quadrati delle cifre che formano il numero precedente.
Es. 4, 16, 37, 58, 89, 145, 42, 20, 4

Osservazioni

- * I numeri che producono sequenze che ritornano sul numero di partenza sono molto rari
Nei primi 1000 numeri ve ne sono 9:
1, 4, 16, 20, 37, 42, 58, 89, 145
- * I numeri che generano sequenze che convergono a 1 sono 142:
7, 10, 13, 19, 23,
- * I numeri che generano sequenze senza fine sono 849:
8, 9, 11, 12, 14,

*Cherry illimitati...
con la brezza
fioriscono gli anemoni*

Programma

```
100 'I numeri Cherry
110 CLS : KEY OFF
120 INPUT "Limite inferiore: ",LI
130 PRINT
140 INPUT "Limite superiore: ",LS
150 DIM CH$(LS)
160 CLS
170 FOR NP=LI TO LS
180   PRINT NP;" ";
190   N=NP
200   C=0
210   N$=STR$(N)
220   L=LEN(N$)
230   FOR K=1 TO L
240     A$=MID$(N$,K,1)
250     A=VAL(A$)
260     C=C+A^2
270   NEXT K
280   PRINT C; : T=T+1
290   IF T>15 AND C <> 1 THEN CH$(NP)="Senza
    fine" : SF=SF+1 : GOTO 330
300   IF T>15 AND C=1 THEN CH$(NP)="Converge a
    1" : CU=CU+1 : GOTO 330
```

```

310 IF C <> NP THEN N=C : GOTO 200
320 CH$(NP)="* CHERRY *" : CH=CH+1
330 T=0
340 PRINT
345 PRINT
350 IF NP/11 = NP\11 THEN LOCATE 23,35 :
    INPUT"Premi invio",D : CLS
360 NEXT NP
370 LOCATE 23,32 : INPUT"Invio per l'analisi ",D : CLS
380 FOR K=LI TO LS
390 PRINT USING"####";K;PRINT TAB(8) CH$(K)
400 IF K\20=K/20 AND K<>LI THEN LOCATE 23,35
    : INPUT"Premi invio ",D : CLS
410 NEXT K
420 LOCATE 23,30 : INPUT"Invio per il riassunto ",D :
    CLS
430 PRINT"Serie illimitata: ";SF
440 PRINT
450 PRINT"Convergono a 1: ";CU
460 PRINT
470 PRINT"CHERRY: ";CH
480 LOCATE 23,1
490 END

```

COMPOSTI - Sono i numeri interi, maggiori di 1 e non primi (vedi *).

Ogni numero composto è scomponibile nel prodotto di numeri primi in un solo modo, a meno dell'ordine (Teorema fondamentale dell'aritmetica)

I primi numeri composti sono: 4 - 6 - 8 - 9 - 10

Osservazioni

- * Lo zero non viene considerato numero composto perché ha infiniti divisori
- * I numeri composti sono infiniti
- * Esistono numeri composti pari e numeri composti dispari

Ad una stella antica

*il bambù si apre..
nascono i Composti*

Programma

```
100 'Scomposizione di un numero Composto
110 CLS : KEY OFF
120 INPUT "Numero da scomporre > 1: ",N
130 IF N<2 THEN GOTO 110
140 A=N
150 DIM E(SQR(N))
160 CLS
170 FOR K=2 TO SQR(N)
180   GOSUB 320
190   IF F=1 THEN 220
200   Q=N/K
210   IF Q=INT(Q) THEN E(K)=E(K)+1 : N=Q : GOTO
200
220 NEXT K
230 PRINT "Scomposizione di";A;": " : PRINT
240 FOR X=2 TO K-1
250   IF E(X)<>0 THEN PRINT X;"^";E(X) : G=1
260 NEXT X
270 IF G=1 AND N<>1 THEN PRINT N
280 IF G=1 THEN PRINT " è composto"
290 IF G=0 THEN PRINT " è primo"
300 LOCATE 23,1
310 END
320 'Generazione dei numeri primi
330 F=0
340 FOR J=K TO SQR(K)
350   Q1=K/J
360   IF Q1=INT(Q1) AND K<>2 THEN F=1 : GOTO
380
370 NEXT J
380 RETURN
```

CONSERVATIVI - Un numero intero è conservativo se
è un multiplo del suo bifronte (vedi *).

Es. $8712:2178=4$

Osservazioni

- * A parte i numeri palindromi (vedi *) o quei numeri che, privati degli zeri finali, sono ancora palindromi, come 1210 o 4400, nei primi 10000 numeri troviamo 17 numeri conservativi:
510, 540, 810, 2100, 4200, 5100, 5200, 5400, 5610, 5700, 5940, 6300, 8100, 8400, 8712, 8910, 9801
- * Solo 8712 e 9801 sono numeri conservativi che non terminano per zero.
- * Entrambi questi numeri sono dei quadrati perfetti e lo sono anche i loro quozienti

*Un raggio splende
con il pesco in fiore
respira il Conservativo*

Programma

```
100 'I numeri Conservativi
110 CLS : KEY OFF
120 INPUT "Limite inferiore: ",A
130 PRINT
140 INPUT "Limite superiore: ",B
150 CLS
160 FOR N=A TO B
170   BF$=""
180   N$=STR$(N)
190   L=LEN(N$)
200   FOR K=L TO 2 STEP -1
210     BF$=BF$+MID$(N$,K,1)
220   NEXT K
230   BF=VAL(BF$)
240   IF BF = N THEN GOTO 290
250   GOSUB 310
260   IF F=1 THEN 290
270   Q=N/BF
280   IF Q=INT(Q) THEN T=T+1 : PRINT T,N;" / ";
      BF;"="";Q
290 NEXT N
300 END
310 'Controllo possibile palindromia
```

```

320 F=0 : NB$=""
330 NN$=STR$(BF)
340 L1=LEN(NN$)
350 FOR J=L1 TO 2 STEP -1
360   NB$=NB$+MID$(NN$,J,1)
370 NEXT J
380 NB=VAL(NB$)
390 IF NB=BF THEN F=1
400 RETURN

```

COPRIMI - Due numeri sono coprimi quando non hanno divisori in comune. Si dice anche che sono primi fra di loro.

Osservazioni

- * Due numeri consecutivi sono coprimi
- * Due numeri dispari consecutivi sono coprimi
- * Se due numeri sono primi (vedi *), sono evidentemente anche primi tra di loro
- * Il massimo comun divisore di due numeri coprimi è 1
- * Il minimo comune multiplo di due numeri coprimi è il loro prodotto

*Fugge la notte...
senza storia
brillano ancora i Coprimi*

Programma

```

100 'Test per i numeri Coprimi
110 CLS : KEY OFF
120 INPUT"Scrivi un numero: ",A
130 PRINT
140 INPUT"Scrivine un altro: ",B
150 M=A : N=B
160 CLS
170 R=1
180 WHILE R <> 0
190   R=A MOD B
200   A=B : B=R
210 WEND

```

```

220 IF A=1 THEN A$="" ELSE A$="non "
230 LOCATE 13,30
240 PRINT M;"e";N;A$;"sono coprimi"
250 LOCATE 23,1
260 END

```

CUBICI - Dal greco kybos=dado, cubo.

Sono quei numeri che rappresentano la terza potenza di un dato numero n , vale a dire n^3 .

Per i greci, infatti, la terza potenza di un numero equivaleva al volume di un cubo con spigolo di uguale misura.

Celebre è il problema della 'duplicazione del cubo':

Gli Ateniesi, colpiti dalla peste, decisero di recarsi a Delo, al tempio di Apollo, portando doni ed un altare a forma cubica. Ma il sacerdote del tempio disse loro che la grandezza di Apollo necessitava di un altare di volume doppio. Così gli Ateniesi, inesperti di matematica, raddoppiarono lo spigolo dell'altare, col risultato di portare al tempio un cubo di volume otto volte più grande.

Infatti se il volume era 1, il sacerdote voleva un cubo di volume 2; ma se lo spigolo era 2, il volume diventava $2^3=8$.

I primi numeri cubici sono: 1 - 8 - 27 - 64 - 125

Osservazioni

- * I numeri cubici sono infiniti
- * Il doppio di un numero cubico non è mai cubico
- * Alcuni numeri cubici sono anche quadrati (vedi *)

*Nel mare di Delo
si specchia il ciliegio
impossibile Cubico...*

Programma

```

100 'I numeri Cubici da 1 a 1000
110 CLS : KEY OFF
120 FOR K=1 TO 10
130   C=K*K*K

```



```

140 PRINT C;
150 Q=SQR(C)
160 IF Q=INT(Q) THEN PRINT " * quadrato" : ELSE
    PRINT
170 NEXT K
180 END

```

DEFICIENTI - Dal latino deficio=manco. Un numero intero è deficiente quando è maggiore della somma dei suoi divisori propri.

SDP(10): $1+2+5=8$; $10>8$; allora 10 è deficiente

Osservazioni

- * Nei primi 1000 numeri ci sono 750 numeri deficienti e, a parte il 945, tutti i numeri dispari sono deficienti
- * I numeri primi (vedi *) sono deficienti
- * Ci sono al massimo 5 numeri deficienti consecutivi

*Tace l'allodola
amata pioggia...
riappare il Deficiente*

Programma

```

100 'I Numeri Deficienti
110 CLS : KEY OFF
120 INPUT "Limite inferiore ";A
130 PRINT
140 INPUT "Limite superiore ";B
150 CLS
160 FOR N=A TO B
170   SD=1
180   FOR K=2 TO SQR(N)
190     Q=N/K
200     IF Q=K THEN SD=SD+K : GOTO 220
210     IF Q=INT(Q) THEN SD=SD+K+Q
220   NEXT K
230   IF N>SD THEN T=T+1 : PRINT T,N
240   IF T/20=T\20 AND C<>T THEN PRINT : INPUT "
    Premi INVIO",D : CLS : C=T
250 NEXT N

```

```

260 PRINT : PRINT"Totale Deficienti: ";T
270 LOCATE 23,1
280 END

```

DISPARI - Sono dispari i numeri non divisibili per 2.
Secondo la Scuola Pitagorica, i dispari e i pari (vedi *)
derivano dal principio del mondo e dall'Unità.

Dalla contrapposizione di questi numeri nascono i
contrasti della Natura. Infatti, il dispari rappresenta il
principio maschile, l'illimitato.

Anche nella cultura cinese il dispari, lo yang,
rappresenta il chiaro, la luce, il principio maschile.

Osservazioni

- * La somma di due numeri dispari è sempre pari.
Infatti: $(2n+1)+(2n+3)=4n+4=4(n+1)$ che è pari
- * La somma di un numero dispari con un numero pari è
dispari.
Infatti: $(2n+1)+2n=4n+1$ che è dispari
- * Il prodotto di due numeri dispari è sempre dispari.
Infatti: $(2n+1)*(2n+3)=4n^2+6n+2n+3=4n^2+8n+3$. I
primi due addendi sono pari, il terzo è dispari, quindi
 $p+p+d=d$
Il prodotto di un numero dispari con un numero pari è
pari. Infatti: $(2n+1)*2n=4n^2+2n$. $p+p=p$
- * E' facile calcolare la somma dei primi n numeri dispari:
La somma dei primi 3 numeri dispari: $1+3+5=9$
Ma $9=3*3$. Quindi per calcolare la somma dei primi n
numeri dispari basta fare $n*n$

*Dispari innocente...
un'orchidea
dolcemente riappare*

Programma

```

100 'I Numeri Dispari e loro somme
110 CLS : KEY OFF
120 INPUT"Limite superiore ",A
130 CLS
140 T=1

```

```

150 FOR K=1 TO A STEP 2
160   PRINT T,A
170   T=T+1
180 NEXT K
190 PRINT"La somma dei primi";T-1;"numeri dispari
    è:";(T-1)^2
200 END

```

DISPARIMENTI PARI - I numeri disparimenti pari, assieme ai parimenti pari e ai parimenti dispari (vedi **), appartengono all'antica classificazione che Nicomaco, matematico greco vissuto nel II secolo a Gerash, non lontano da Gerusalemme, fece dei numeri pari (vedi *). Sono quei numeri che si possono suddividere, anche più volte, in parti uguali, ma non fino all'unità.

Es. $12=6+6$; $6=3+3$

I numeri disparimenti pari si ottengono moltiplicando fra di loro, in tutti i modi possibili, i numeri parimenti pari con i numeri dispari.

Sono generati dalla formula $2^m \cdot (2n+1)$ con $m \geq 2$ ed $n \geq 1$

Osservazioni

- * Tutti i numeri disparimenti pari sono divisibili per 4
- * Nei primi 1000 numeri naturali i disparimenti pari sono 241. Vale a dire, tutti i numeri divisibili per 4, che sono 250, escludendo le potenze del 2 che sono 9

*Nel cielo di Gerash
si riposano i Disparimenti pari
con la luna del deserto*

Programma

```

100 'I numeri Disparimenti pari fino a 1000
110 CLS : KEY OFF
120 DIM V(1000)
130 FOR N=2 TO 8
140   P2=2^N
150   FOR D=3 TO 249 STEP 2
160     DP=P2*D

```



```

170     IF DP <= 1000 THEN V(DP)=DP
180  NEXT D
190 NEXT N
200 T=1
210 FOR N=1 TO 1000
220   IF V(N) <> 0 THEN PRINT T,V(N) : ELSE GOTO
      250
230   IF T/20 = T\20 THEN INPUT C : CLS
240   T=T+1
250 NEXT N
260 END

```

DUFFYNIANI - Dall'inglese to duff=sbagliare il colpo.
 Un numero intero positivo è duffyniano se la somma dei
 suoi divisori propri non è divisibile per i suoi addendi
 (con 1 escluso)
 SDP(50): $1+2+5+10+25=43$; ma 2, 5, 10, 25 non sono
 divisori di 43 allora 50 è duffyniano

Osservazioni

- * Nei primi 1000 numeri ci sono 233 numeri duffyniani
- * I numeri primi, la cui somma dei divisori propri è 1,
sono numeri duffyniani 'banali'
- * I numeri perfetti (vedi *) non sono duffyniani.
- * I numeri quasiperfetti (vedi *), cioè le potenze del 2,
sono tutti duffyniani

*Il sole dorato si risveglia
 Duffyniano:
 verde palude che tace*

Programma

```

100 'I numeri Duffyniani
110 DIM D(100)
120 CLS : KEY OFF
130 INPUT "Limite inferiore: ",A
140 PRINT
150 INPUT "Limite superiore: ",B
160 CLS
170 FOR N=A TO B

```

```

180 SD=1 : F=0
190 FOR K=2 TO SQR(N)
200   Q=N/K
210   IF Q=K THEN SD=SD+K : D(K)=K : GOTO 230
220   IF Q=INT(Q) THEN SD=SD+K+Q : D(K)=K :
      D(Q)=Q
230 NEXT K
240 IF SD=1 THEN 320
250 FOR J=2 TO N/2
260   IF D(J)=0 THEN 290
270   IF SD/D(J)=INT(SD/D(J)) THEN F=1
280   D(J)=0
290 NEXT J
300 IF F=0 THEN T=T+1 : PRINT T,N : C=N
310 IF T/20=T\20 AND F=0 THEN LOCATE 23,5 :
      INPUT"Premi invio ",D : CLS
320 NEXT N
330 LOCATE 23,1
340 END

```

ETEROGENEI - Dal greco éteros=diverso e
génos=specie

Sono quei numeri interi i cui fattori primi hanno tutti
l'esponente uguale a 1

FP(42): 2, 3, 7 allora 42 è eterogeneo

Osservazioni

- * Nei primi 1000 numeri ci sono 439 eterogenei più 168
numeri primi che sono, evidentemente, eterogenei
- * Tutti i numeri eterogenei non hanno la radice quadrata
la loro 'irrazionalità' è quindi irriducibile
- * Esistono più numeri eterogenei che omogenei (vedi *)

*Il picchio riluce
Eterogenei...
Nitida vallata che scorre*

Programma

```

100 'I numeri Eterogenei
110 CLS:KEY OFF

```

```

120 INPUT "Limite inferiore: ",A
130 PRINT
140 INPUT "Limite superiore: ",B
150 CLS
160 FOR N=A TO B
170   P=N
180   GOSUB 330
190   IF F=0 THEN GOTO 300
200   FOR T=2 TO SQR(N)
210     O=0 : P=T
220     GOSUB 330
230     IF F=1 THEN 260
240     Q=N/T
250     IF Q/T=INT(Q/T) THEN O=1 : GOTO 270
260   NEXT T
270   IF O=0 THEN SE=SE+1 : PRINT SE,N
280   IF SE=CE THEN 300
290   IF SE/20=SE\20 AND SE<>0 THEN LOCATE 23,5
      : INPUT "Premi invio ",D : CE=SE : CLS
300 NEXT N
310 LOCATE 23,1
320 END
330 'Test dei numeri primi
340 F=0 : ND=0
350 FOR K=2 TO SQR(P)
360   Q=P/K
370   IF Q=INT(Q) THEN ND=ND+2
380   IF ND>1 THEN F=1 : GOTO 400
390 NEXT K
400 RETURN

```

ETEROLETTERALI - Dal greco éteros=diverso e dal latino litterale=lettera.

Sono quei numerali (vedi *) il cui nome è composto da lettere tutte diverse.

Osservazioni

* Esperanto: nulo, du, tri, kvar, kvin, sep, ok, nau, dek, dektri, deknau, tridek, trideknau, naudek, naudektri, cent, ..., ducent, kvarcent, okcent,

mil, ..., dekmil

- * Francese: zéro, un, deux, trois, quatre, cinq, six, sept, huit, neuf, dix, onze, douze, quatorze, quinze, dix-sept, dix-neuf, vingt, soixante, cent
- * Inglese: zero, one, two, four, five, six, eight, ten, thousand
- * Italiano: zero, uno, due, tre, cinque, sei, nove, venti, cento, centomila
- * Spagnolo: cero, uno, dos, tres, cuatro, diez, once, doce, quince, ochenta, cien, mil
- * Tedesco: eins, zwei, drei, vier, acht, zehn, elf, zwolf, sechzig, achtzig, hundert, tausend
- * I numeri eteroletterali non sono infiniti
- * Delle lingue esaminate, l'Esperanto, sembra avere il maggior numero di eteroletterali

*Oggi la farfalla
è immobile come una foglia
Eteroletterale lontano*

FATTORIALI - Il fattoriale di un numero è il prodotto di tutti i numeri da 1 a n.

Es. $6! = 1 \cdot 2 \cdot 3 \cdot 4 \cdot 5 \cdot 6 = 720$

Osservazioni

- * Per convenzione $0! = 1$; $1! = 1$
- * Tutti i fattoriali, ad eccezione di 0 e 1, sono pari e, a partire dal 5, terminano tutti per zero
- * La successione dei fattoriali tende a crescere molto velocemente. $18!$ è il numero massimo che il nostro elaboratore può calcolare
- * I fattoriali sono utili per calcolare il numero delle permutazioni di n elementi. Ad es., il numero di anagrammi teorici della parola 'Informatica' è $11! = 39.916.800$ 'Minor fatica' e 'Fratelli in coma', sono alcuni di questi

*Fattoriale notturno
alla roccia
una camelia riposa*

Programma

```
100 'I Fattoriali di un numero
110 CLS : KEY OFF
120 INPUT "Scrivi un numero < 19: ",N
130 IF N >18 THEN 110
140 CLS
150 PRINT 0; : PRINT TAB(22);1
160 F#=1
170 FOR K=1 TO N
180   F#=F#*K
190   PRINT USING "##";K;
200   PRINT USING "#####";F#
210 NEXT K
220 LOCATE 23,1
230 END
```

FIBONACCI - da 'Filius Bonacci', soprannome di Leonardo Pisano, matematico italiano vissuto nel XIII secolo e famoso per aver introdotto in Italia le cifre indo-arabiche.

Sono quei numeri della serie dove, a parte i primi due, ogni elemento è la somma dei due che lo precedono.

I primi numeri di Fibonacci: 1, 1, 2, 3, 5, 8, 13, 21,....

Osservazioni

- *I numeri di Fibonacci sono utili per risolvere il problema dei conigli: quante coppie di conigli si ottengono in un anno, supponendo che ogni coppia dia alla luce un'altra coppia ogni mese e che, quest'ultima, dal secondo mese in poi, sia in grado di riprodursi? 1 coppia in gennaio, 2 in febbraio, 3 in marzo, 5 in aprile, 8 in maggio, 13 in giugno, 21 in luglio, 34 in agosto, 55 in settembre, 89 in ottobre, 144 in novembre, e, finalmente, 233 in dicembre.
- *Il 10° numero di Fibonacci, 55, rappresenta, anche, la somma dei primi 10 numeri naturali
- * Il 12° è 144 che è il quadrato di 12. Nei primi 50 numeri di Fibonacci, questo è l'unico quadrato perfetto
- * Nella successione, il rapporto fra un numero e il suo successivo tende alla 'sezione aurea' 0,618...che é il formato di questo volume

*Come in un sogno un lampo
svanisce nel nulla -
Fibonacci misterioso*

Programma
100 '1 numeri di Fibonacci
110 CLS : KEY OFF
120 INPUT "Limite superiore <51: ",A
130 IF A >50 THEN 110
140 CLS
150 F1#=1 : F2#=1
160 PRINT 1; : PRINT TAB(20) F1#
170 PRINT 2; : PRINT TAB(20) F2#
180 FOR K=3 TO A
190 FIB#=F2#+F1#
200 PRINT USING "###";K;
210 PRINT USING "#####";FIB#;
220 IF K/20 = K\20 AND K<>A THEN LOCATE 23,5 :
INPUT "Premi invio ",D : CLS
230 F1#=F2# : F2#=FIB#
240 NEXT K
250 LOCATE 23,1
260 END

GUINESSIANI - Sono quei numeri che, a vario titolo,
sono apparsi nel Guinness dei primati. (anno 1990)

Osservazioni

- * Nel sistema decimale, il numero più grande al quale corrisponde un nome è il 'Centillion', cioè $(10^6)^{100}$
Al di fuori del sistema decimale, il numero più grande è 'Asankhyeya' buddista, vale a dire 10^{140}
- * Il numero più grande scomposto in fattori è un numero di 100 cifre. Lo si è ottenuto grazie alla collaborazione di 400 calcolatori che hanno lavorato per 26 giorni

*Presso il tramonto
l'alba chiara trascorre
gelido Guinnessiano*

HOTEL - Sono quei numeri che vengono usati negli alberghi per contrassegnare le stanze.

Osservazioni

- * La prima cifra o le prime due cifre rappresentano il numero del piano, le altre il numero della stanza
- * Sono un esempio rarissimo di numeri formati da numeri ordinali, le prime cifre, e, numeri stringa (vedi *) le ultime
- * E' difficile trovare numeri hotel le cui prime o ultime cifre siano 13 o 17

*Una memoria invernale
risveglia il mattino
fiorisce l'Hotel*

INTERI - dal latino integr=integro, intero. Sono quei numeri che esprimono una o più volte l'unità.

Osservazioni

- * E' famoso l'aneddoto che racconta come il giovane Gauss, a soli 10 anni, fosse riuscito a trovare la formula per calcolare la somma dei primi n numeri naturali che, come è noto, è: $SN=(n+1)*n/2$
- * In informatica il termine 'integer' indica il tipo di variabile che corrisponde, in matematica, al termine intero sia positivo che negativo, ma con una limitazione: le variabili possono assumere valori negativi e positivi entro un intervallo, 'range', che dipende dal particolare tipo di elaboratore.
Nei nostri personal l'intervallo è -32768 e +32767

Programma

```
100 'La somma dei numeri Interi
110 CLS : KEY OFF
120 INPUT"Limite superiore: ",N
130 CLS
140 LOCATE 13,23
150 PRINT"La somma dei primi";N;" numeri è:
    ";(1+N)*N/2
```

160 LOCATE 23,1
170 END

*Il sole illumina la foresta:
tra le brume
splendono gl'Interi*

LOTTO - Sono quei numeri che vengono estratti ogni sabato per la gioia e, a volte, il dolore dei giocatori.

Osservazioni

- * I numeri del lotto sono i numeri naturali da 1 a 90
- * Vengono premiati, di ogni ruota, il numero singolo, l'ambo, il terno, la quaterna, la cinquina
- * Esiste una ricchissima letteratura sulla corrispondenza fra sogno e numeri del lotto.
Ad esempio, 4 è il papà, 17 l'amico, 47 il morto che parla, 52 la mamma, 90 la paura,

*Oggi un raggio
splende con il pesco
il Lotto è gelato*

Programma

```
100 'Estrazioni del Lotto
110 CLS : KEY OFF
120 DIM F(90)
130 INPUT "Quante estrazioni: ",Q
140 CLS
150 RANDOMIZE TIMER
160 FOR N=1 TO Q
170   CLS
180   C=0
190   LOCATE 2,25: PRINT "BARI"
200   LOCATE 4,21: PRINT "CAGLIARI"
210   LOCATE 6,22: PRINT "FIRENZE"
220   LOCATE 8,23: PRINT "GENOVA"
230   LOCATE 10,23: PRINT "MILANO"
240   LOCATE 12,23: PRINT "NAPOLI"
250   LOCATE 14,22: PRINT "PALERMO"
```

```

260 LOCATE 16,25: PRINT "ROMA"
270 LOCATE 18,23: PRINT "TORINO"
280 LOCATE 20,22: PRINT "VENEZIA"
290 RI=2:CO=31
300 WHILE C<>10
310     LOCATE RI,CO
320     FOR K=1 TO 5
330         E=INT(RND*90+1)
340         IF F(E)=0 THEN F(E)=E : PRINT USING"#####";
            E; : PRINT " "; : ELSE GOTO 330
350         C=C+1:RI=RI+2
360     NEXT K
370 WEND
380 FOR V=1 TO 90 : F(V)=0 : NEXT
390 LOCATE 23,33
400 IF N=Q THEN PRINT"F I N E" : GOTO 440
410 PRINT"giocata numero ";N;
420 X$=INKEY$
430 IF X$ <> CHR$(13) THEN 420
440 NEXT N
450 LOCATE 23,1
460 END

```

MAGICI - Il Numero, dicevano i Pitagorici, è la legge dell' Universo e come tale è l'essenza di tutte le cose. Noi non sappiamo come siano nati i numeri: forse per un'esigenza pratica, o forse anche perché erano collegati ad antichissimi rituali magici. Oggi, molti di questi rituali sono andati perduti e spesso quel che rimane è il significato magico legato a particolari numeri.

ZERO - (vedi *)

UNO - (vedi *)

DUE - è il primo numero pari ed è collegato all'idea di coppia e di instabilità che per i Pitagorici è il simbolo dell'imperfezione e della femminilità. Gli antichi non lo consideravano un numero fortunato, come del resto non lo erano tutti i numeri che iniziavano con la cifra 2. I Romani dedicavano il secondo mese dell'anno a Plutone, il dio degli inferi, e nel secondo giorno dello

stesso mese facevano sacrifici ai morti.

TRE - da sempre, il tre, è collegato all'idea di divinità. Le trinità, infatti, sono molte nella storia delle religioni. Horus, Isisde, Osiride per gli Egiziani; Brahma, Siva, Visnu per gli indiani; Padre, Figliolo, Spirito Santo; e Minerva, Giunone, Venere le tre dee della Grecia classica. E ancora tre sono le Parche le Furie, le Grazie, le Arpie, i Re Magi.

E' il numero della perfetta armonia, dunque simbolo maschile.

QUATTRO - è il numero della stabilità, dell'equilibrio fisico e della giustizia. Ancora oggi, in Inghilterra, per indicare un uomo giusto si dice 'a square man'.

CINQUE - siccome nasce dall'unione del primo numero dispari (maschile) con il primo numero pari (femminile) era considerato il numero del matrimonio.

SEI - è il primo numero perfetto, ottenuto sia dalla somma che dal prodotto dei suoi divisori propri. Il venerdì (sesto giorno della settimana) è ancor oggi considerato un giorno sfortunato.

SETTE - è sacro e misterioso e corrisponde al numero dei pianeti. E' il numero della saggezza: sette, infatti erano i Saggi dell'antichità classica. E nella religione cristiana sono sette i doni dello Spirito Santo, i Sacramenti, le Virtù e i peccati capitali. La pianta della Moschea è a forma eptagonale e il candelabro della Sinagoga ha sette bracci. E ancora sette sono i giorni della settimana, le note musicali e i nani di Biancaneve

OTTO - era, per i Greci, il simbolo dell'amore e dell'amicizia, rappresentazione dell'uguaglianza tra gli uomini.

NOVE - è detto il numero del lutto perché, presso i Romani, il periodo del lutto durava nove giorni. Ancor oggi i funerali di un Pontefice si svolgono nove giorni dopo la sua morte.

I Greci consigliavano di evitare il 9 e il suo

quadrato 81 ($8+1=9$).

DIECI - per i Greci era il numero ideale e rappresentava l'Universo perché si ottiene dalla somma dei primi quattro numeri: 1 rappresenta il punto, 2 la linea, 3 il piano e 4 il tetraedro (3 punti di base e 1 vertice della piramide). Sempre dai Greci, veniva invocato come testimone nei loro giuramenti.

*In un canneto nebbioso
mormora ancora il Magico
vento di terre lontane...*

NARCISISTI - Sono quei numeri il cui valore è uguale alla somma dei cubi delle loro cifre.

Es. $153=1^3+5^3+3^3$

Il nome prende spunto dal mito greco di Eco e Narciso. Narciso era un giovane che, vedendo la sua immagine riflessa nell'acqua, se ne innamora ed Eco era una ninfa innamorata, purtroppo senza successo, di Narciso.

Osservazioni

- * Nei primi 1000 numeri ci sono solo cinque narcisisti:
1, 153, 370, 371, 407
- * Se un numero non è narcisista possiamo, di nuovo, calcolare la somma dei cubi delle cifre e, continuando con questo sistema, ottenere due possibili esiti:
la successione incontra, mediamente dopo 10 passaggi, un dei quattro numeri narcisisti e si ferma.
Es. 152 ; 134 ; 92 ; 737 ; 713 ; 371 narcisista d'ordine 5.
Nei primi 1000 numeri ve ne sono in tutto 875
la successione entra in un cerchio senza fine.
Es. 55 ; 250 ; 133 ; 55. Questi sono i narcisisti illimitati e sono 117
- * Fra i narcisi illimitati ci sono due numeri la cui somma dei cubi delle loro cifre è uguale all'altro numero e viceversa. 136 genera il 244 e 244 genera il 136; 919 genera il 1459 e 1459 genera il 919.
Li abbiamo chiamati Eco e Narciso

*A un fantasma nitido
il tempo sorride.
Eco e Narciso*

Programma

```
100 'I numeri Narcisisti
110 CLS : KEY OFF
120 INPUT "Scrivi un numero: ",N
130 CLS
140 S=0
150 N$=STR$(N)
160 L=LEN(N$)
170 FOR K=1 TO L
180   C$=MID$(N$,K,1)
190   C=VAL(C$)
200   P=C*C*C
210   S=S+P
220 NEXT K
230 J=J+1
240 PRINT USING"###";J;:PRINT TAB(10)
    USING"####";N;:PRINT TAB(20) USING"####";S
250 IF J=20 THEN LOCATE 20,30: PRINT "Serie
    illimitata" : GOTO 270
260 IF N<>S THEN N=S : GOTO 140
270 LOCATE 23,1
280 END
```

NUMERALI - Si chiamano numerali i nomi dei numeri scritti in lettere. Numerale, per es., è la parola che usiamo quando compiliamo un assegno (si dice anche il numero scritto in lettere).
(vedi anche ETEROLETTERALI e ONESTI)

Osservazioni

- * In genere quando si parla di numero si intendono le cifre che lo rappresentano; è perciò difficile pensare ai numerali come ad una categoria di numeri.
- * In questo insieme è soprattutto difficile ordinare alfabeticamente i 'numeri', in quanto bisogna far riferimento a un campo che, di solito, consideriamo

- lontano dalla matematica: le lettere dell'alfabeto
- * Tra i numerali e i naturali c'è una corrispondenza biunivoca: ad ogni numero corrisponde un numerale e viceversa. Entrambi gli insiemi sono infiniti.
 - * Per rappresentare i numerali conviene utilizzare una metafora: un albero dal quale partono infiniti rami (generatori) i quali, a loro volta, producono altri infiniti rami (elementi secondari).
 GENERATORI: sono quei numerali di cui non si può decidere chi sia il predecessore e che generano una serie infinita di successori. Es. CINQUE, SEI, DODICI, VENTI
 ELEMENTI SECONDARI: sono gli infiniti derivati dei generatori.
 Es. SEImila, VENTItre, TREcento
 - * Nella lingua italiana, l'insieme dei numerali, pur essendo infinito, ha un inizio (CENTO) e una fine (ZERO).
 Ma questo è solo un caso: in altre lingue, ad esempio il tedesco, il primo numero è ACHT (otto) e l'ultimo non esiste. Si può solo identificare l'ultimo generatore: ZWOLF (dodici).
 - * Se, per esempio, volessimo ordinare i numeri da 0 a 10 otterremmo questa successione:
 CINQUE, DIECI, DUE, NOVE, OTTO, QUATTRO, SEI, SETTE, TRE, UNO, ZERO
 Ma se dilatassimo l'intervallo, per esempio fino a venti, la nostra successione non sarebbe più la stessa.

*A lungo il torrente
 scorre con il tempo
 limpidi Numerali...*

OMAR - Vengono chiamati numeri di Omar perché chi li ha proposti è lo studioso Omar Calabrese.
 I numeri della successione si ottengono in questo modo: dato il primo numero, 1, il successivo è la descrizione del precedente: 11, e di nuovo il successivo sarà 21. Le cifre di ogni successione sono di due tipi: 'sostantivi' e 'aggettivi'.

Le cifre 'sostantivo' indicano il numero vero e proprio mentre le cifre 'aggettivo' indicano la quantità del sostantivo.

I primi dieci numeri di Omar:

1
11
21
1211
111221
312211
13112221
1113213211
31131211131221
13211311123113112211

Osservazioni

- * I numeri di Omar sono infiniti
- * Tutti, tranne il primo, hanno un numero di cifre pari
- * Sono costituiti solo dai primi tre numeri naturali (1,2,3)
- * Tutti i numeri di Omar finiscono per 1
- * Le cifre aggettivo sono numericamente uguali alle cifre sostantivo
- * Non abbiamo trovato nessuna legge che ci calcoli il numero delle cifre dell'ennesimo numero di Omar

*Un nebbia marina trascorre
palpita l'Omar...
luna invernale*

Programma

```
100 'I numeri di Omar
110 DIM A$(256) , B$(256)
120 RI=2 : CO=5
130 CLS : KEY OFF
140 INPUT "Primo numero di Omar: ",A$
150 CLS
160 PRINT 1 ; " " ; A$ ; : PRINT TAB(77) ; 1
170 FOR Q=1 TO 17
180   T=1
190   FOR K=1 TO LEN(A$)
```

```

200   A$(K) = MID$(A$,K,1)
210  NEXT K
220  FOR M=1 TO LEN(A$)
230    FOR J=M+1 TO LEN(A$)
240      IF A$(M) = A$(J) THEN T=T+1
      ELSE GOTO 260
250    NEXT J
260    IF T=0 THEN B$ = B$ + A$(M) : GOTO 290
270    T$=STR$(T)
280    B$ = B$ + RIGHT(T$,1) + A$(M)
290    M=J-1 : T=1
300  NEXT M
310  FOR S=1 TO LEN(B$)
320    B$(S) = MID$(B$,S,1)
330  NEXT S
340  PRINT USING"###" ; Q+1
350  LOCATE RI,CO
360  FOR S1=1 TO LEN(B$)
370    IF S1=70 OR S1=139 THEN
      RI=RI+1:CO=5:LOCATE RI,CO
380    PRINT B$(S1);
390  NEXT S1
400  PRINT TAB(76) USING"###";LEN(B$)
410  RI=RI+1
420  A$=B$
430  B$=""
440 NEXT Q
450 LOCATE 23,1

```

OMOGENEI - Dal greco *homòs*=uguale e *génos*=specie
 Un numero intero è omogeneo se almeno un suo fattore
 primo ha l'esponente maggiore di 1
 FP(60): $2^2 \cdot 3 \cdot 5$ allora 60 è omogeneo

Osservazioni

- * Nei primi 1000 numeri ci sono 392 omogenei
- * Esistono 11 triplete di omogenei
- * Al massimo ogni quattro numeri c'è un omogeneo
- * I quadrati perfetti e i loro multipli sono omogenei

*Oggi un'orchidea viola
si muove nella pianura
freme l'Omogeneo...*

Programma

```
100 'I numeri Omogenei
110 CLS : KEY OFF
120 INPUT"Limite inferiore: ",A
130 PRINT
140 INPUT"Limite superiore: ",B
150 CLS
160 FOR N=A TO B
170   P=N
180   GOSUB 330
190   IF F=0 THEN GOTO 300
200   FOR T=2 TO SQR(N)
210     O=0 : P=T
220     GOSUB 330
230     IF F=1 THEN 260
240     Q=N/T
250     IF Q/T=INT(Q/T) THEN O=1 : GOTO 270
260   NEXT T
270   IF O=1 THEN SO=SO+1 : PRINT SO,N
280   IF SO=CO THEN 300
290   IF SO/20=SO\20 AND SO<>0 THEN LOCATE
      23,5 : INPUT"Premi invio ",D : CLS : CO=SO
300 NEXT N
310 LOCATE 23,1
320 END
330 'Test dei numeri primi
340 F=0 : ND=0
350 FOR K=2 TO SQR(P)
360   Q=P/K
370   IF Q=INT(Q) THEN ND=ND+2
380   IF ND>1 THEN F=1 : GOTO 400
390 NEXT K
400 RETURN
```

ONESTI - Sono quei numerali (vedi *) che sono composti da tante lettere quante ne indica il numero

stesso.

Osservazioni

- * 1 - A (Inglese antico), I (Cinese)
- 2 - CA (Maya), DA (Irlandese), DU (Esperanto), NI (Giapponese), TO (Danimarca)
- 3 - ESH (Sumero), SAN (Cino-Giapponese), TRE (Italiano, Spagnolo, Danese, Norvegese), TRI (Sanscrito, Russo, Serbo-croato, Esperanto), YEY (Azteco)
- 4 - DORT (Turco), FIRE (Danese, Norvegese), FOUR (Inglese), FYRA (Svedese), KVAR (Esperanto), NAVI (Azteco), VIER (Tedesco)
- 5 - CHICA (Azteco), CINCI (Rumeno), CINCO (Portoghese, Spagnolo), FEMMA (Svedese), PANCA SANSCRITO, PEEZI (Lettone), PENKI (Lituano), PENTE (Greco), PIATI (Polonia), VIISI (Finlandia)
- 6 - CHWECH (Gallese)
- 7 - CHICOME (Azteco), SEPTYNI (Lituano)
- 8 - ATTETALL (Norvegese)
- 9 - KOKONOTSU (Giapponese)
- 10 - DZIESIATKA (Polacco)
- 11 - NJIMBEDHIET (Albanese)
- 12 - UMIKUMAMALUA (Hawaiano)
- 13 - UMIKUMAMAKOLU (Hawaiano)
- 14 - TESSARAKAIDEKA (Greco)
- 15 - ISHUMI NESIHLANU (Zulu)
- 16 - KUUSTEISTKUMMEND (Estone)
- * Attualmente i ricercatori hanno trovato i numeri onesti da 1 a 16, ma, nei 5800 diversi linguaggi stimati nel mondo, è possibile trovarne altri?
- * Lo ZERO è disonesto in tutte le lingue

*La nuvola svanisce
palpita l'Onesto...
luminoso mattino*

PALINDROMI - Dal greco palin=all'indietro e dromos=corsa.

Sono quei numeri che letti da sinistra a destra e da destra a sinistra, non cambiano il loro valore.

Es. 7 - 11 - 121 - 12521

La palindromia è nascosta in ogni numero, ma essa non vuol rivelare compiutamente il suo segreto.

Infatti, se un numero non è palindromo si somma ad esso il suo bifrante (vedi *) e ripetendo più volte la stessa operazione, si arriva quasi sempre ad ottenere un numero palindromo.

Es. $158 + 851 = 1009$

$1009 + 9001 = 10010$

$10010 + 01001 = 11011$ palindromo di ordine 3

Negli interi fino a 200, la maggior parte dei numeri diventa palindromo al massimo in 4 passaggi.

I più ostinati, assieme ai loro bifronti, sono: 166/176/198 (5 passaggi) - 79/182 (6 p.) - 188/190/197 (7 p.) - 193 (8 p.) - 167 (11 p.) - 177 (15 p.) - 187 (23 p.) - 89 (24 p.) - 196 (?? p.)

Osservazioni

- * Un numero diventa palindromo nello stesso numero di passaggi del suo bifrante
- * Se la somma delle cifre di un numero è < di 10 allora il numero è palindromo di ordine 1
- * Se la somma delle cifre di un numero è > di 10 e anche palindroma, allora il numero è palindromo di ordine 1
- * Nei primi 1000 numeri naturali il 196, il 295, il 394 e i loro bifronti non hanno ancora mostrato il loro palindromo
- * Con il 196, due studiosi americani hanno provato a fare 237.310 addizioni senza ottenere alcun risultato
- * Tutti i numeri con cifre dispari, divisibili per 11, sono palindromi
- * Nell'insieme dei numerali (vedi *), nella lingua italiana, solo l'OTTO è palindromo, in Esperanto abbiamo UNU e SES
- * Oltre ai numeri sono interessanti anche le date palindrome. Se consideriamo solo l'anno, il 1991 è palindromo.

La data più recente risale al 28/9/1982 e la prossima

sarà il 19/9/1991. E' facile osservare che nel secolo XX le date palindrome saranno solo nel mese di Settembre; nel XIX secolo solo nel mese di Agosto e così via fino all'XI secolo. Inoltre, tutti gli anni che terminano con cifre superiori al 3 non possono avere date palindrome

*Il mare estivo tace
Palindromo...
lontana isola che dorme*

Programma

```
100 'Test Palindromia dei numeri
110 CLS : KEY OFF
120 US$="#####"
130 INPUT"Scrivi un numero ",N#
140 CLS
150 N$=STR$(N#)
160 B$=""
170 L=LEN(N$)
180 FOR K=L TO 1 STEP -1
190   B$=B$+MID$(N$,K,1)
200 NEXT K
210 BF#=VAL(B$)
220 IF T=0 AND N#=BF# THEN PRINT N#;" E' già
    palindromo!" : GOTO 320
230 WHILE N#<>BF#
240   M#=N#+BF#
250   T=T+1
260   PRINT USING"##";T;;PRINT USING US$;
    N#;;PRINT" +";BF#;TAB(40)" = ";M#
270   IF T/20=T\20 THEN PRINT:INPUT"   PREMI
    INVIO", D : CLS
280   IF T=33 THEN PRINT:PRINT TAB(27)"Numero
    troppo grande!" : GOTO 320
290   N#=M#
300   GOTO 150
310 WEND
320 LOCATE 23,1
330 END
```

PARI - Sono pari i numeri divisibili per 2.

Secondo la Scuola Pitagorica, i pari e i dispari (vedi *) derivano dal principio del mondo e dall'Unità. Dalla contrapposizione di questi numeri nascono i contrasti della Natura. Infatti, il pari rappresenta il principio femminile, il limitato. Anche nella cultura cinese il pari, lo yin, rappresenta l'oscurità, le tenebre, l'ombra, il principio femminile.

Osservazioni

- * La somma di due numeri pari è sempre pari.
Infatti: $2n+2n=4n$ che è pari
- * La somma di un numero pari con un numero dispari è dispari. Infatti: $2n+(2n+1)=4n+1$ che è dispari.
- * Il prodotto di due numeri pari è sempre pari.
Infatti: $2n*(2n+2)=4n^2+4n=4n(n+1)$ che è pari
- * Il prodotto di un numero pari per uno dispari è dispari
Infatti: $2n(2n+1)=$ pari
- * E' facile calcolare la somma dei primi n numeri pari
Somma dei primi 3 numeri pari: $2 + 4 + 6 = 12$
Ma $12 = 3 * 4$. Quindi per calcolare la somma dei primi n numeri pari basta fare $n * (n+1)$
- * Nicomaco, matematico greco vissuto nel II secolo a Gerash, non lontano da Gerusalemme, divide i numeri pari in parimenti pari, parimenti dispari, disparimenti pari (vedi * * *)

*Pari oscuro
nella notte silenziosa
la luna sussurra*

Programma

```

100 'Numeri Pari e loro somme
110 CLS : KEY OFF
120 INPUT "Limite Superiore: ",A
130 CLS
140 T=1
150 FOR K=2 TO A STEP 2
160   PRINT T,K
170   T=T+1
180 NEXT K

```

```

190 PRINT"La somma dei primi";T-1;"numeri pari è:
    ";(T-1)*T
200 END

```

PARIMENTI DISPARI - I numeri parimenti dispari, assieme ai parimenti pari e ai disparimenti pari, (vedi *), appartengono all'antica classificazione che Nicomaco, matematico greco vissuto nel II secolo a Gerash, non lontano da Gerusalemme, fece dei numeri pari (vedi *). Sono quei numeri divisibili una sola volta in parti uguali. Sono, quindi, il doppio di un numero dispari, e si ottengono con la formula $2*(2n+1)$ con $n \geq 1$

Osservazioni

- * I numeri parimenti dispari costituiscono una successione di ragione 4: 6 ; 10 ; 14 ; 18 ; 22 ; 26 ;
- * Considerando un numero pari di termini, la somma di ogni coppia di numeri equidistanti è costante
Es. $6 + 26 = 10 + 22 = 14 + 18$
Se il numero dei termini è dispari, la somma dei termini equidistanti è uguale al doppio del termine centrale
Es. $6 + 22 = 10 + 18 = 2 * 14$

*La luna di Gerash si leva
respirano i Parimenti dispari
solitaria peonia*

Programma

```

100 'I numeri Parimenti dispari finì a 1000
110 CLS : KEY OFF
120 FOR N=3 TO 999 STEP 2
130   PD=2 * N
140   PRINT PD
150   IF K/41 = K\41 THEN PRINT : INPUT D : CLS
160 NEXT N
170 END

```

PARIMENTI PARI - I numeri parimenti pari, assieme ai parimenti dispari e ai disparimenti pari, (vedi *), appartengono all'antica classificazione che Nicomaco,

matematico greco vissuto nel II secolo a Gerash, non lontano da Gerusalemme, fece dei numeri pari (vedi *). Sono i numeri 'più pari' di tutti, quelli divisibili per 2 fino all'unità. In pratica, sono parimenti pari tutte le potenze del 2 ottenute dalla formula 2^n con $n \geq 1$

Osservazioni

- * Nei primi 1000 numeri, ci sono solo 9 parimenti pari
- * I numeri parimenti pari costituiscono una progressione geometrica di ragione 2: 2 ; 4 ; 8 ; 16 ; 32 ; ...
- * La somma dei suoi termini è sempre uguale al successivo diminuito di un'unità
- * Considerando un numero pari di termini, il prodotto di ogni coppia di numeri equidistanti è costante
Es. $1 \cdot 32 = 2 \cdot 16 = 4 \cdot 8$
- * Se il numero dei termini è dispari, il prodotto dei i termini equidistanti è uguale al quadrato del termine centrale
Es. $1 \cdot 16 = 2 \cdot 8 = 4^2 = 16$

*Nel tramonto di Gerash
riappaiono i Parimenti pari
dolce eco...*

Programma

```
100 'I numeri Parimenti pari fino a 1000
110 CLS : KEY OFF
120 FOR N=1 TO 9
130   PP=2^N
140   PRINT N+1,PP
150 NEXT N
160 END
```

PERFETTI - Un numero intero positivo è perfetto quando è uguale alla somma dei suoi divisori propri. Il più piccolo numero perfetto è $6=1+2+3$. La storia di questi numeri è misteriosa e per questo affascinante! Ci sono molte leggende in proposito: Sant'Agostino afferma addirittura che Dio ha creato il mondo in 6 giorni e non in un minuto perché il 6 è

perfetto.

E la Luna gira intorno alla Terra in 28 giorni!

Euclide fu il primo a trovare la formula dei numeri perfetti, formula che è legata ai numeri primi di Mersenne.

Scriva Euclide: "Se $2^n - 1$ genera un numero primo allora $2^{(n-1)} * (2^n - 1) = \text{numero perfetto}$ ".

I primi 4 numeri perfetti erano già noti nell'antichità.

Me(2)	$2^1 * (2^2 - 1)$	6
Me(3)	$2^2 * (2^3 - 1)$	28
Me(5)	$2^4 * (2^5 - 1)$	496
Me(7)	$2^6 * (2^7 - 1)$	8.128
Me(13)		33.550.336
Me(17)		8.589.869.056
Me(19)		137.438.691.328
Me(31)		2.305.843.008.139.952.128
Me(61)	$2.658.455.991.569.831.744.654.692.615.953.842.176$	

Con l'uso del computer, si sono trovati trenta numeri primi di Mersenne, il più grande dei quali ha 65050 cifre. Dunque $2^{216090} * (2^{216091} - 1)$ è il più grande numero perfetto finora conosciuto.

Osservazioni

- * I numeri perfetti conosciuti sono tutti pari e terminano o per 6 o per 8.
- * I numeri perfetti (escluso il 6) sono sempre divisibili per quattro, congrui a 4 (mod 12) e congrui a 1 (mod 9)
- * Le somme dei reciproci di tutti i divisori di un numero perfetto è 2
Es. 28 $1 + 1/2 + 1/4 + 1/7 + 1/14 + 1/28 = 2$
Da qui deriva un'altra possibile definizione:
Se la somma dei reciproci di tutti i divisori di n è uguale a 2, allora n è perfetto.
- * Tutti i numeri perfetti sono triangolari (vedi *)
- * E' facile scrivere un numero perfetto in forma binaria.
Della formula euclidea il primo esponente rappresenta il numero degli zero e il secondo il numero degli uno.
Es. $2^4 * (2^5 - 1) = 496 = 111\ 110\ 000$ (base 2)

*Perfetto amato
oltre il tramonto
la montagna riluce*

Programma
100 'I numeri Perfetti fino a 10.000
110 CLS : KEY OFF
120 FOR N=2 TO 10000 STEP 2
130 SD=1
140 FOR K=2 TO SQR(N)
150 Q=N/K
160 IF Q=K THEN SD=SD+K : GOTO 180
170 IF Q=INT(Q) THEN SD=SD+K+Q
180 NEXT K
190 IF SD=N THEN PRINT USING"####";N; : PRINT"
è Perfetto"
200 NEXT N
210 END

PITAGORICI - I numeri pitagorici, meglio conosciuti come terne pitagoriche, sono quei numeri, in ordine crescente, dove la somma dei quadrati dei primi due è uguale al quadrato del terzo.

La più famosa e la più antica è 3, 4, 5, perché
 $3^2+4^2=5^2$

Osservazioni

- * Pitagora fu il primo a trovare una formula che generasse le terne: b , $(b^2-1)/2$, $(b^2+1)/2$ con b dispari

La formula aveva il pregio di generare terne primitive, terne, cioè, che non avessero divisori in comune, ma non generava tutte le possibili. Le terne Pitagoriche si riconoscono perché l'ipotenusa e il cateto maggiore differiscono di 1

- * Platone ci provò con un'altra formula:

b , $(b^2-4)/4$, $(b^2+4)/4$ con b pari

Con questa formula si ottengono terne che Pitagora non aveva scoperto, come per es. 8, 15, 17, ma si ottengono anche tante terne non primitive come 6, 8,

10 che deriva dalla terna primitiva 3, 4, 5.

Nelle terne Platoniche l'ipotenusa e il cateto maggiore differiscono di 2

- * Toccò a Euclide, ma altri studiosi dicono Diofanto, a creare una formula generale:
dati m e n, coprими (vedi *) e di parità opposta la terna è:
 $a=m^2-n^2$ $b=2*m*n$ $c=m^2+n^2$
Nelle terne Euclidee la differenza fra l'ipotenusa e il cateto maggiore non è costante: minimo è 8
- * Uno dei primi due numeri di una terna primitiva è sempre pari e il terzo è sempre dispari
- * Un numero della terna è sempre divisibile per 3 e un altro per 5 e il prodotto di tutti e tre è divisibile per 60

*La memoria invernale respira
 Pitagorici...
vecchio orizzonte che mormora*

Programma

```
100 'Le terne Pitagoriche, Platoniche, Euclidee
110 CLS : KEY OFF
120 LOCATE 8,20 :PRINT"Terne Pitagoriche 1"
130 LOCATE 10,20 :PRINT"Terne Platoniche 2"
140 LOCATE 12,20 :PRINT"Terne Euclidee 3"
150 LOCATE 14,20 :PRINT"FINE LAVORO 4"
160 LOCATE 18,30 :PRINT"Scegli 1 / 2 / 3 / 4";
170 X$=INKEY$ : X=VAL(X$)
180 IF X<1 OR X>4 OR X=0 THEN GOTO 170
190 IF X=4 THEN LOCATE 23,1 : END
200 ON X GOSUB 220, 400, 590
210 GOTO 110
220 'Terne Pitagoriche
230 CLS
240 PRINT TAB(26) "LE PRIME 10 TERNE
    PITAGORICHE"
250 RI=3 : CO=30
260 FOR N=3 TO 21 STEP 2
270   LOCATE RI,CO
280   A=N
290   B=(N^2-1)/2
```

```

300 C=(N^2+1)/2
310 PRINT USING"##";A;
320 PRINT USING"#####";B;
330 PRINT USING"#####";C
340 RI=RI+2
350 NEXT N
360 LOCATE 25,26 : PRINT"< INVIO PER TORNARE
    AL MENU >";
370 X$=INKEY$
380 IF X$<>CHR$(13) THEN 370
390 RETURN
400 'Terne Platoniche
410 CLS
420 PRINT TAB(26)"LE PRIME 10 TERNE
    PLATONICHE"
430 RI=3 : CO=30
440 FOR M=8 TO 46 STEP 2
450   LOCATE RI,CO
460   A=M
470   B=(M^2-4)/4
480   C=(M^2+4)/4
490   IF A/2=A\2 AND B/2=B\2 THEN 540
500   PRINT USING"##";A;
510   PRINT USING"#####";B;
520   PRINT USING"#####";C
530   RI=RI+2
540 NEXT M
550 LOCATE 25,26 : PRINT"< INVIO PER TORNARE
    AL MENU >";
560 X$=INKEY$
570 IF X$<>CHR$(13) THEN 560
580 RETURN
590 'Terne Euclidee
600 CLS
610 PRINT TAB(30)"ALCUNE TERNE EUCLIDEE"
620 RI=3 : CO=30
630 FOR N=2 TO 7
640   FOR M=N+1 TO 7
650     R=1
660     IF N/2<>N\2 AND M/2<>M\2 THEN M=M+1

```

```

670     X=M : Y=N
680     WHILE R <> 0
690         R=X MOD Y
700         X=Y : Y=R
710     WEND
720     IF X<>1 THEN 820
730     A=M*M-N*N
740     B=2*M*N
750     C=M*M+N*N
760     LOCATE RI,CO
770     IF A>B THEN SWAP A,B
780     PRINT USING"###";A;
790     PRINT USING"#####";B;
800     PRINT USING"#####";C
810     RI=RI+2
820 NEXT M
830 NEXT N
840 LOCATE 25,26 : PRINT"< INVIO PER TORNARE
      AL MENU >";
850 X$=INKEY$
860 IF X$<>CHR$(13) THEN 850
870 RETURN

```

PLANICI - Dal latino planus=piano. Sono quei numeri che si ottengono sommando a un quadrato la sua radice.

Es. $1^2+1=2$; $2^2+2=6$

Sono, quindi, generati dalla formula $n^2 + n$

Osservazioni

- * Siccome $n^2+n=n*(n+1)$, i numeri planici si possono ottenere, anche, moltiplicando un numero per il suo successivo $2*3=6$; $4*5=20$
- * Ma, moltiplicando un numero per il suo successivo si ottiene anche la somma dei primi n numeri pari, quindi i numeri planici coincidono con tali somme, e sono tutti pari
Es. $2+4+6=3*4=12$
- * Ricordando che $n/2*(n + 1)$ è la formula che dà la somma dei primi n numeri naturali, i numeri planici si possono ottenere anche raddoppiando tale somma.

$$1+2+3=6 \text{ e } 3^2+3=12$$

*L'orchidea fiorisce
sorride il Planico
umida ninfea*

Programma

```
100 'I numeri Planici fino a 1000
110 CLS : KEY OFF
120 FOR N=1 TO 1000
130   PL=N^2+N
140   PRINT PL
150 NEXT N
160 END
```

PRIMI - Un numero intero, maggiore di 1, è primo se è divisibile solo per sé stesso e uno.

DP(23)= 23, 1, quindi 23 è primo

Osservazioni

- * I numeri primi sono infiniti. La dimostrazione la dobbiamo a Euclide. Nei numeri da 2 a 1000 ci sono 168 numeri primi
- * Il più grande numero primo, oggi conosciuto, è esprimibile con la formula $2^{216091} - 1$.
E' un numero di 65.050 cifre
- * A tutt'oggi, invece, non si conosce ancora la formula con la quale poter trovare tutti e solo i numeri primi
- * A parte il 2 e il 5, tutti gli altri numeri primi terminano per 1, 3, 7, 9
- * Nell'insieme dei numeri primi sono stati costruiti molti particolari sottoinsiemi:
 - Bifronti: numeri che letti da sinistra a destra e da destra a sinistra sono ancora primi. Es. 13, 31
Nei primi 1000 numeri ce ne sono 18
 - Gemelli: numeri primi la cui differenza è 2. Es. 17, 19
Nei primi 1000 numeri ci sono 35 coppie
 - Palindromi: numeri che letti da sinistra a destra e da destra a sinistra hanno lo stesso valore. Es 131
Nei primi 10000 numeri ce ne sono 20

*Una nebbia svanisce
Primo...
antico, oscuro torrente*

```
Programma
100 'I numeri Primi
110 CLS : KEY OFF
120 INPUT "Limite inferiore >1: ",A
130 IF A<2 THEN GOTO 110
140 PRINT
150 INPUT "Limite superiore: ",B
160 CLS
170 FOR N=A TO B
180   F=0
190   FOR J=2 TO SQR(N)
200     Q=N/J
210     IF Q=INT(Q) THEN F=1 : J=SQR(N)
220   NEXT J
230   IF F=0 THEN T=T+1:PRINT USING "#####";T;N
240   IF T/20=T\20 AND C<>T THEN LOCATE 23,2 :
      INPUT "Premi invio ",D : CLS
250   C=T
260 NEXT N
270 LOCATE 23,1
280 END
```

QUADRATI - Sono quei numeri che rappresentano il risultato di un numero moltiplicato per se stesso.

Es. $4=2^2$

Per i greci, infatti, un numero elevato alla seconda equivaleva all'area di un quadrato di lato uguale al numero dato.

I primi numeri quadrati sono: 1 - 4 - 9 - 16 - 25

Osservazioni

- * I numeri quadrati sono infiniti
- * Il doppio di un numero quadrato non è mai quadrato
- * Alcuni numeri quadrati sono anche cubici (vedi *)

Presso una profonda notte

*una piccola luce si spegne
il Quadrato sussurra...*

Programma

```
100 'I numeri Quadrati da 1 a 1000
110 CLS : KEY OFF
120 FOR K=1 TO 31
130   Q=K*K
140   PRINT Q;
150   C=Q^(1/3)
160   IF C=INT(C) THEN PRINT " * cubico" : ELSE
       PRINT
170 NEXT K
180 END
```

QUASIPERFETTI - Un numero intero positivo si dice quasiperfetto quando è maggiore o minore di 1 della somma dei suoi divisori propri.

SDP(8): $1+2+4=7$; $8-7=1$ allora 8 è quasi perfetto

Osservazioni

- * Nei primi 1000 numeri ci sono 9 numeri quasiperfetti
- * Finora non si conoscono numeri quasiperfetti con la somma dei divisori maggiore di 1, né si è dimostrato che non ne possano esistere
- * I numeri quasiperfetti, con la somma dei divisori minore di 1, sono solo le potenze del 2

*Il campo bianco
in silenzio -
i Quasiperfetti*

Programma

```
100 'I numeri Quasiperfetti fino a 1000
110 CLS : KEY OFF
120 FOR N=2 TO 1000
130   SD=1
140   FOR K=2 TO SQR(N)
150     Q=N/K
160     IF Q=K THEN SD=SD+K : GOTO 180
```



```

170     IF Q=INT(Q) THEN SD=SD+K+Q
180  NEXT K
190  IF N=SD+1 OR N=SD-1 THEN T=T+1 : PRINT
      T,N
200 NEXT N
210 PRINT"Totale quasiperfetti: ";T
220 END

```

ROMANI - I numeri romani sono forse l'ultimo esempio di numerazione additiva usato ancor oggi, se non altro, per indicare i secoli e rappresentare gli ordinali.

I Romani, per rappresentare un numero, si servivano di sette simboli, le cifre romane, appunto.

I (uno), V (cinque), X (dieci), L (cinquanta), C (cento), D (cinquecento), M (mille)

Le regole erano queste:

- a) Nessuna cifra può essere ripetuta consecutivamente per più di tre volte. Erano, evidentemente, consapevoli del fatto che il conteggio diventa difficile quando un simbolo si ripete per più di quattro volte.
- b) Una cifra, se è collocata a destra di una di maggior valore, viene sommata. Es. VI=V+I=6
- c) Una cifra, se è collocata a sinistra di una di maggior valore, viene sottratta. Es. XL=X-L=40
- d) Una linea orizzontale posta sopra la cifra la rende 1000 volte più grande e 2 linee 1.000.000 di volte più grande.

Osservazioni

- * Considerando le cifre romane come delle lettere di un mini-alfabeto si possono comporre delle parole e calcolare il loro valore.

VI+LI=6+51=57 , LI+VI+DI=51+6+501=557 ,

VI+VI=6+6=12

*In una foresta equatoriale
svanisce il Romano.
solitaria cascata...*

SOCIEVOLI - I numeri socievoli sono legati al concetto

di numeri amici (vedi *) dove la catena che porta al numero originario è di due anelli. Ma, se la catena ne ha più di due, i numeri corrispondenti sono detti socievoli. La somma dei divisori propri di 12.496 = 14.288, quella di 14.288 = 15.472 ; di 15.472 = 14.536 e di 14.536 = 14.264. Infine la somma dei divisori di 14.264 = 12.496

Catena di 4 anelli:

$SD(1.264.468)=1.547.860$; $SD(1.547.860)=$
 $1.727.636$; $SD(1.727.636)=1.305.184$;
 $SD(1.305.184)=1.264.468$

Altri 12 numeri che formano catene di 4 anelli:

2.115.324 - 2.784.580 - 4.938.136 - 7.169.104
 18.048.976 - 18.656.380 - 46.722.700 - 81.128.632
 174.277.820 - 209.524.210 - 330.003.580 -
 498.215.416

Catena di 28 anelli: 14.316 ; 19.116 ; 31.704 ; 47.616

83.328 ; 177.792 ; 295.488 ; 629.072
 589.786 ; 294.896 ; 358.336 ; 418.904
 366.556 ; 274.924 ; 275.444 ; 243.760
 376.736 ; 381.028 ; 285.778 ; 152.990
 122.410 ; 97.946 ; 48.976 ; 45.946
 22.976 ; 22.744 ; 19.916 ; 17.716 ; 14.316

Osservazioni

- * Le catene che noi conosciamo sono tutte fatte di numeri pari
- * Se il numero degli anelli è 4 allora metà sono abbondanti (vedi *) e metà sono deficienti (vedi *). Nelle altre catene prevalgono i numeri deficienti
- * Non si conoscono catene di tre anelli, ma nessuno ha dimostrato che non ne possano esistere.

*Una pineta fredda sorride
 si aprono i Socievoli
 brillante tramonto*

Programma

100 'Test per i numeri Socievoli
 110 CLS : KEY OFF
 120 INPUT N

```

130 CLS
140 G=1
150 PRINT G; : PRINT USING"#####";N
160 A=N
170 SD=1
180 FOR K=2 TO SQR(N)
190   Q=N/K
200   IF Q=INT(Q) THEN SD=SD+K+Q
210 NEXT K
220 IF G/20=G\20 THEN INPUT D : CLS
230 IF G > 30 OR SD=1 THEN PRINT"Non ho trovato i
    soci di "; A : END
240 IF A=SD THEN END : ELSE N=SD : G=G+1 : PRINT
    G; : PRINT USING"#####";SD : GOTO 170

```

STRINGA - Sono quei numeri che, come i nomi delle persone, vengono usati solo per identificare oggetti diversi.

L'autobus 37 va sul percorso 37 e il 91 va sul percorso 91.

Osservazioni

- * Con i numeri stringa si può eseguire un'unica operazione: uguale (=) e non uguale (<>)
- * In informatica si può eseguire anche l'operazione di concatenazione (+). "veli" + "volo" = velivolo
- * Anche nei numerali esistono numeri stringa: la Fiat UNO
- * I numeri delle camere degli hotel (vedi *) sono stringhe
- * I numeri stringa hanno generato le stringhe alfanumeriche, vale a dire, stringhe composte da numeri e lettere: il dentifricio AZ15, la vitamina B12, il codice fiscale...

*In un canneto nebbioso
mormora la Stringa
eterna notte.*

SUPERABBONDANTI - Un numero è superabbondante se è la metà della somma dei suoi divisori propri.
SDP(120)=240 , $240/120=2$ allora 120 è Superabbondante

Osservazioni

- * Nei primi 1000 numeri ce ne sono solo due: 120, 672 e sono entrambi pari
- * La metà di un superabbondante è abbondante (vedi *)
- * I multipli dei superabbondanti sono abbondanti

*Superabbondanti amati
oltre la pioggia...
il campo sognato*

Programma

```
100 'I numeri Superabbondanti fino a 1000
110 CLS : KEY OFF
120 FOR N=2 TO 1000 STEP 2
130   SD=1
140   FOR K=2 TO SQR(N)
150     Q=N/K
160     IF Q=K THEN SD=SD+K : GOTO 180
170     IF Q=INT(Q) THEN SD=SD+K+Q
180   NEXT K
190   IF T/20=T\20 THEN INPUT D : CLS
200   IF N=SD/2 THEN T=T+1 : PRINT T,N
210 NEXT N
220 PRINT "Totale superabbondanti: ";T
230 END
```

TRIANGOLARI - Sono quei numeri rappresentabili con una serie di punti, disposti in modo da formare un triangolo equilatero. Lo studio di queste successioni è dovuto alla scuola Pitagorica (V secolo) che se ne servì per analizzare importanti leggi dell'aritmetica. (Vedi anche QUADRATI).

I primi numeri triangolari sono: 1, 3, 6, 10, 15

Sono ottenuti dalla somma dei primi n numeri naturali con la formula $n/2 * (n+1)$

Il numero 10 era un numero triangolare molto venerato dai Pitagorici perché era formato dalla somma dei primi 4 numeri naturali e il triangolo ottenuto lo chiamavano 'Tetractis' dal greco tetra=quattro

Osservazioni

- * I triangolari sono, alternativamente, due dispari e due pari
- * I numeri perfetti (vedi *) sono triangolari

*L'anemone si mostra
Triangolari...
le onde s'infrangono*

Programma

```
100 'I numeri Triangolari
110 CLS : KEY OFF
120 INPUT "Quanti numeri triangolari ";Q
130 CLS
140 FOR N=1 TO Q
150   T=N/2*(N+1)
160   PRINT USING"## ";N, : PRINT USING"
      #####";T
170 NEXT N
180 LOCATE 23,1
190 END
```

UNO - E' per i Greci l'unico numero parimpari, cioè né pari né dispari, che può trasformare i numeri pari in dispari e viceversa. E' anche il numero che, con addizioni ripetute, produce tutti gli altri. Per questo motivo, era il simbolo della Divinità e della Ragione. In aritmetica, l'uno ha un ruolo particolare che dipende dalle situazioni in cui si presenta. Infatti, nel mondo della divisibilità viene rigorosamente escluso. Non è né primo né composto per non contraddire il teorema fondamentale dell'aritmetica che dice:

'un numero è scomponibile nel prodotto dei suoi fattori primi in un'unica maniera'.

Viene, invece, recuperato, con degli assiomi, tutte le volte che si presentano delle operazioni un po' particolari.

Es. $a^0=1$ con $a \neq 0$; $0!=1$

Uno, Uno, Uno

*nell'impossibile foresta
la successione non ha fine*

VENTI - Dal latino 'viginti', è il numero cardinale che indica una quantità composta da due decine.

Anticamente,alcune tribù, che contavano servendosi anche delle dita dei piedi, lo hanno utilizzato come base del loro sistema di numerazione. Gli Aztechi dividevano il giorno in 20 ore e il loro esercito in reparti di 8000 soldati ($8000=20*20*20$).

Nella lingua francese il numero 80 (quatre-vingt) rappresenta la spia di un possibile antico sistema di numerazione in base 20. La stessa cosa accade nella lingua inglese con i termini score (20), two-score (40), three-score (60)

*Nel cielo dei tropici
mani e piedi
aspettano il Venti*

ZERO - Lo zero è il numero che indica il numero degli elementi di un insieme vuoto.

Partendo dallo zero si possono generare tutti gli altri numeri naturali. I Greci e i Romani non utilizzavano lo zero nel loro sistema di numerazione. I Babilonesi utilizzavano uno spazio vuoto, generando, però, molta confusione.

In un tempio non lontano da Gvalior, cittadina dell'India centrale, troviamo la prima iscrizione dove viene rappresentato lo zero. L'iscrizione risale all'870 d.c.

Gli indiani chiamavano lo zero, 'sunya'=vuoto.

Nel IX secolo, Abu Jafar Muhammad Ibn Musa al Khwarismi (Mohammed, padre di Jafar e figlio di Mosè, della provincia persiana di Khoresm) pubblicò un libro in cui spiegava il sistema di numerazione indiano, dando allo zero il nome di 'sifr'=vuoto.

Nel 1202, Leonardo Pisano, più conosciuto come Fibonacci (vedi *), pubblica il 'Liber Abaci' dove introduce, per la prima volta in Europa, il sistema di numerazione decimale e propone di chiamare lo zero,

'zephirum'. L'affermazione dello zero, però, arriverà in Europa molto più tardi, nel 1500.

In aritmetica lo zero pone molte sottili questioni:

La definizione classica di moltiplicazione va in crisi con $a * 0$, ed anche quella di potenza con a^0 .

Nell'addizione in N è elemento neutro. Nella moltiplicazione è elemento assorbente, ma nella divisione è elemento..... impotente!. Ha infiniti divisori, e proprio per questo viene escluso dal mondo della divisibilità.

Insomma, lo zero, è proprio un cifra scomoda e i matematici, quando si trovano in difficoltà, lo eliminano letteralmente dall'insieme. Infatti, spesso si parla di N^* , Z^* , Q^* , privati cioè, dello zero.

*Da sempre lo Zero
è segno della mancanza
ma anche mancanza di segno*

Firmano il lavoro con
tautogrammi ropalici simmetrici

AVEVA
ANGELI
ALLEGRI
ALLUNATI
AGILMENTE
ALESSANDRA
AMICABILI
AFFAMATI
ABITANO
ANCORA
ABETI

AVVIA
ADESSO
ATOMICI
ACROBATI
ACCLAMATI
ALESSANDRO
ABILMENTE
ANIMERAI
ATTRICI
ATTORI
ACERI

ANCHE
ACERBA
ALESSIA
ADOTTAVA
AMOREVOLI
ARCHITETTI
AUTENTICI
ARROTINI
ANSIOSI
AGENTI
ACIDI

AMICI
ADATTI
ALLERTA
ANNALISA
ASCOLTAVA
ABBONDANTI
ADDIZIONI
AGITANDO
AZZURRI
AIRONI
ALATI

ARAVI
ANDREA
ASSENTI
AGRUMETI
ARGENTATI
ACCOGLIEVI
ABILMENTE
AQUILONI
APATICI
ANANAS
ARABI

AGILI
ARAZZI
ANTONIO
ANALIZZA
ANNODANDO
ARTICOLATI
AUTOCARRI
AFFILATI
AUTISTI
ASTICI
AVIDI

ANIMA
ANGELI
ARIANNA
AFFRESCA
ANALITICI
ARCOBALENI
ARMONICHE
AUSTRALI
ASCISSE
AFFINI
AROMI

CERCO
CLASSI
CURIOSI
CREATIVE
CEREBRALI
COORDINATE
CAMBIANDO
CREATURE
CONFUSO
CESARE
CORRE

COGLI
CHIARA
CAMELIE
CUBITALI
CONGELATE
CAUTAMENTE
CAVALIERI
COMMOSSI
COPRONO
CACTUS
COTTI

ELENA
ELETTA
EREDITA
ENNESIMO
ERMELLINO
ECCELLENTE
ECOLOGICO
ERBIVORO
ERETICO
EPPURE
EBETE

SUPER
SILVIA
SCARICA
SCAFFALI
SOLAMENTE
SCOLASTICI
STUDIANDO
SVARIATI
SISTEMI
SLALOM
SFUSI

SENTI
SIMONE
SBADATO
SOPPRIMI
STRATEGIE
SILENZIOSE
SMASCHERA
SAPIENTI
SCALTRI
SEMPRE
SAGGI

SIAMO
SERENI
SICCOME
STEFANIA
SOTTOVOCE
SOCIALIZZA
SPLENDIDA
STUDIOSA
SVANITA
SEMPRE
SETTE

Mantova, giugno 1991